

Uso do simulador elétrico SPICE

O simulador elétrico SPICE permite a simulação de circuitos eletrônicos descritos textualmente, onde estão inseridos os comandos de descrição do circuito e de análise requerida:

Para criação do arquivo texto de descrição é importante seguir algumas regras básicas, respeitando a sintaxe de descrição de cada componente (elementos passivos, ativos e fontes) assim como a sintaxe dos comandos que devem ser usados para uma determinada simulação.

Regras - Resumo

Regra 1 – Primeira linha do arquivo sempre é usada como título, e não tem influencia nenhuma na simulação. Usado normalmente para identificar a descrição do arquivo. Um comando importante colocado na primeira linha será desconsiderado e usado como título do arquivo.

Regra 2 – Unidades aceitas pelo SPICE e seus símbolos:

t	10e+12	u	10e-06
g	10e+09	n	10e-09
meg	10e+06	p	10e-12
k	10e+03	f	10e-15
m	10e-03		

Obs.: pode-se usar também a letra ‘e’ para o expoente na base 10, ao invés da unidade, por exemplo: 4e-12 = 4p.

Regra 3 – Os nós ou nodos do circuito a serem descritos devem ser nomeados ou numerados (*a numeração é preferencial*) antes de descrever cada elemento.

Regra 4 – O número **0** (zero) sempre é usado para indicar o terra, massa ou *ground* do circuito. É o ponto referencial para as medidas de tensão.

Regra 5 – A descrição de fontes de tensão sempre são nomeadas começando pela letra ‘V’ e de fontes de corrente pela letra ‘I’. Seguem algumas exemplos de fontes:

DC – Vnome nó_1(+) nó_2(-) valor_dc.

Ex.: VCC 1 0 5 (fonte de tensão entre nós 1 (+) e 0 (-) de 5V)

Senoidal – Vnome nó_1(+) nó_2(-) sin(offset amplitude frequência tempo_início)

Quadrada – Vnome nó_1(+) nó_2(-) pulse(tensão_1 tensão_2 início_pulso tempo_subida tempo_descida largura _pulso período)

Linear (quadrada/triangular) – Vnome nó1(+) nó2(-) pwl(tempo_1 tensão_1 tempo_2 tensão_2 tempo_3 tensão_3 ... tempo_n tensão_n)




Figura 1 - Exemplo de sinais de tensão DC e Quadrada.

Regra 6 – Descrição de elementos passivos:

Resistor – Rnome nó1(+) nó2(-) valor_resistência
Capacitor – Cnome nó1(+) nó2(-) valor_capacitância
Indutor - Lnome nó1(+) nó2(-) valor_indutância

Obs.: Nunca devem haver dois elementos com mesmo nome. Sugestão: numerar os elementos como R1, R2, C1, C2, C3,...

Regra 7 – Descrição de elementos ativos. Os elementos ativos sempre exigem a descrição do modelo tecnológico com os parâmetros de processo de fabricação. Se os parâmetros não estão descritos os valores default (padrões) são assumidos.

Diodo – Dnome nó1(+) nó2(-) nome_modelo
Transistor Bipolar – Qnome coletor base emissor nome_modelo
Transistor MOS – Mnome dreno gate fonte substrato nome_modelo (w=? l=?)

Regra 8 – Comando para análise transiente. Indica a taxa de aquisição e o tempo de simulação.

.trans passo_aquisição tempo_final

Regra 9 – Obtenção das tensões nos nós, da análise transiente. Os nós desejados devem ser listados.

.print tran V(nó) V(nó) ...

Regra 10 – Indicação de fim de arquivo.

.end

Regra 11 – Toda a linha de comentário deve começar por '*'. As linhas de comentário não são consideradas pelo simulador.

Execução do programa SPICE

Para executar o SPICE em modo batch, é necessário abrir uma janela de comandos DOS, executando o *command.exe*, de preferência no diretório onde estão os arquivos de trabalho.

O comando a ser executado é o spice3 com opção -b (batch):

```
C:\diretório> spice3 -b arquivo_descricao > arquivo_saida
```

Para ver o gráfico com os sinais simulados, use janela de comandos DOS ou executar o **grafer** diretamente do Windows.

```
C:\diretório> grafer arquivo_saida
```

ou

```
C:\diretório> grafer (File -> Open -> arquivo_saida)
```

Arquivo de Descrição – Resumo

A criação do arquivo de descrição do circuito para simulação pode seguir uma ordem de forma a manter a clareza do arquivo. Sugere-se:

- 1) Título
- 2) Descrição das fontes
- 3) Descrição dos elementos passivos
- 4) Descrição dos modelos ativos
- 5) Comando para chamada dos modelos (**.model**
- 6) Comando para definição do tempo de simulação (**.tran**
- 7) Comando para definição das tensões dos nós a serem fornecidas (**.print tran**
- 8) Comando final (**.end**)
- 9) Comentários podem ser colocados em qualquer linha do programa, iniciando por '*'.

Exemplos

Exemplo 1:

```
Titulo: Circuito Resistivo Simples
* fontes
vcc 1 0 8.5
* circuito
r1 1 2 560
r2 2 0 1k
r3 2 3 330
r4 3 0 470
* comandos
.tran 0.1m 100m
.print tran v(1) v(2) v(3)
.end
```

Exemplo 2:

```
Titulo: Circuito RC Diferenciador e Integrador
vin 1 0 pulse( 0 10 0 10u 10u 1m 02m)
* ..... pulse(v1 v2 TD TR TF PW PER)
* integrador
r1 1 2 2.2k
c1 2 0 80n
* diferenciador
r2 3 0 2.2k
c2 1 3 80n
.tran 0.001m 5m
.print tran v(1) v(2) v(3)
.end
```

Exemplo 3:

```
Titulo: Circuito Logica de Diodos (AP04)
* fontes de tensao
vcc 1 0 dc 5
vin1 10 0 pulse(0 5 0 0 0 10n 20n)
vin2 20 0 pulse(0 5 5n 0 0 10n 20n)
vin3 30 0 pulse(0 5 0 0 0 20n 40n)
vin4 40 0 pulse(0 5 0 0 0 40n 80n)
r1 1 2 100
r2 3 0 1k
d1 2 10 diodo
d2 2 20 diodo
d3 2 30 diodo
d4 40 3 diodo
d5 2 3 diodo
.model diodo d
.tran 0.01n 100n
.print tran v(2) v(3) v(10) v(20) v(30) v(40)
.end
```